

```

-module(sendmory).
-author('patai@lit.bme.hu, hanak@inf.bme.hu').
-vsn('2010-11-27').
-compile(export_all).

%   S E N D
% + M O R E
% = M O N E Y

% @type d() = integer(). % digit values have to be in the range 0..9.
% @type octet() = {d(),d(),d(),d(),d(),d(),d(),d()}.

% @spec num(Ns::[d()]) -> N::integer().
% The value of the digit list Ns, interpreted as a decimal number, is N.
num(Ns)->
    lists:foldl(fun(X,E) -> E*10+X end, 0, Ns).

% @spec smm0() -> [octet()].
% All the checks are after the generators (generate and test).
smm0() ->
    Ds = lists:seq(0, 9),
    [{S,E,N,D,M,O,R,Y} ||
     S <- Ds, E <- Ds, N <- Ds, D <- Ds,
     M <- Ds, O <- Ds, R <- Ds, Y <- Ds,
     all_different([S,E,N,D,M,O,R,Y]),
     S > 0, M > 0,
     begin Send = num([S,E,N,D]),
         More = num([M,O,R,E]),
         Money = num([M,O,N,E,Y]),
         Send+More == Money
     end].

% @spec all_different(Xs::[any()]) -> B::bool()
% B is true, if there is no repeated value in the list Xs.
all_different([]) ->
    true;
all_different([X|Xs]) ->
    not lists:member(X,Xs) andalso all_different(Xs).

% all_different(L) -> length(L) == length(lists:usort(L)).

% @spec smm1() -> [octet()].
% Checking inequalities on the way.
smm1() ->
    Ds = lists:seq(0, 9),
    [{S,E,N,D,M,O,R,Y} ||
     S <- Ds,
     E <- Ds, E /= S,
     N <- Ds, not lists:member(N, [S,E]),
     D <- Ds, not lists:member(D, [S,E,N]),
     M <- Ds, not lists:member(M, [S,E,N,D]),
     O <- Ds, not lists:member(O, [S,E,N,D,M]),
     R <- Ds, not lists:member(R, [S,E,N,D,M,O]),
     Y <- Ds, not lists:member(Y, [S,E,N,D,M,O,R]),
     S > 0, M > 0,
     begin Send = num([S,E,N,D]),
         More = num([M,O,R,E]),
         Money = num([M,O,N,E,Y]),
         Send+More == Money
     end].

```

```

% @spec perm() -> [octet()].
% Generating permutations while excluding digits that are already in use.
perm() ->
    Ns = lists:seq(0,9),
    [{S,E,N,D,M,O,R,Y} ||
     S <- Ns -- [0], % Zero (0) excluded.
     E <- Ns -- [S],
     N <- Ns -- [S,E],
     D <- Ns -- [S,E,N],
     M <- Ns -- [0,S,E,N,D], % Zero (0) excluded.
     O <- Ns -- [S,E,N,D,M],
     R <- Ns -- [S,E,N,D,M,O],
     Y <- Ns -- [S,E,N,D,M,O,R]
    ].

% @spec check(octet()) -> bool().
% Checking the rule.
check({S,E,N,D,M,O,R,Y}) ->
    (1000*S + 100*E + 10*N + D) + (1000*M + 100*O + 10*R + E) ==
    (10000*M + 1000*O + 100*N + 10*E + Y).

% @spec smm2() -> [octet()].
% First generate, then test.
smm2() ->
    lists:filter(fun check/1, perm()).

% @spec smm3() -> [octet()].
% First generate, then test.
smm3() ->
    Ns = lists:seq(0,9),
    [{S,E,N,D,M,O,R,Y} ||
     S <- Ns -- [0],
     E <- Ns -- [S],
     N <- Ns -- [S,E],
     D <- Ns -- [S,E,N],
     M <- Ns -- [0,S,E,N,D],
     O <- Ns -- [S,E,N,D,M],
     R <- Ns -- [S,E,N,D,M,O],
     Y <- Ns -- [S,E,N,D,M,O,R],
     (1000*S + 100*E + 10*N + D) + (1000*M + 100*O + 10*R + E) ==
     (10000*M + 1000*O + 100*N + 10*E + Y)
    ].

% @spec smm4() -> [octet()].
% Handling inequalities by construction (which is actually slower...).
smm4() ->
    [{S,E,N,D,M,O,R,Y} ||
     {[S,E,N,D,M,O,R,Y],_} <- select_n(8, lists:seq(0, 9)),
     S > 0, M > 0,
     begin Send = num([S,E,N,D]),
         More = num([M,O,R,E]),
         Money = num([M,O,N,E,Y]),
         Send+More == Money
     end].

```

